

Inversive Geometry: Part 1

Inverting Generalized Circles, Ellipses, Polygons, and Tilings

Jaime Rangel-Mondragon

This article explores the basic properties of inversive geometry from a computational point of view. Topics included in this part are involutions, generalized circles, and the inversion of segments, arcs, triangles, and quadrilaterals. The applications are to Nicomachus's theorem, the inversion of tilings made by regular polygons, and an inversive spirograph.

■ Introduction

The author of [1] describes the technique of inversion as a “dark art.” A suitable interpretation of this description is offered by [2] as “an advanced technique, which can offer considerable advantage in solving certain problems.” This article examines the basic properties of inversive geometry, starting from the introduction of involutions and the family of generalized circles, the inversion of segments, arcs, triangles, and quadrilaterals with applications to Nicomachus's theorem, the inversion of tilings made by regular polygons, and an inversive spirograph. This work extends and complements the material in [3, 4, 5].

■ Involutions

A transformation T that is not the identity with the property that $T(T(z)) = z$ is called an *involution* (or an *involutory transformation*, *self-inverse*, or *of period two*). Familiar examples of involutions are multiplication by -1 and taking a reciprocal in arithmetic, taking a complement in set theory, taking the conjugate of a complex number (i.e., $\overline{x + iy} = x - iy$, if x and y are real), geometrical reflection in a line, the matrix transpose and matrix inverse, and the mapping of a number x into $a - x$, with a an arbitrary number.

An important example of an involution involves permutations. A permutation under the operation of group composition is an involution if it is of order 2; that is, if it can be written as a product of disjoint transpositions (cycles of length 2). For example, the 7-permutation 5234176 maps $1 \rightarrow 5$, $2 \rightarrow 2$, $3 \rightarrow 3$, $4 \rightarrow 4$, $5 \rightarrow 1$, $6 \rightarrow 7$, $7 \rightarrow 6$; it factors into cycles as (15)(2)(3)(4)(67) or, leaving out the fixed points 2, 3, 4, as (15)(67).

```
PermutationCycles[{5, 2, 3, 4, 1, 7, 6}]
```

```
Cycles[{{1, 5}, {6, 7}}]
```

The number of involutory n -permutations grows quickly with n .

```
ParallelTable[
  Length[Select[Permutations[Range[n]],
    PermutationOrder[#] == 2 &]], {n, 8}]
```

```
{0, 1, 3, 9, 25, 75, 231, 763}
```

These numbers can also be readily obtained by solving a recurrence equation [6].

```
Module[{n},
  RecurrenceTable[
    {a[1] == 0, a[2] == 1, a[n] == a[n - 1] + (1 + a[n - 2]) (n - 1)},
    a[n], {n, 20}]]
```

```
{0, 1, 3, 9, 25, 75, 231, 763, 2619, 9495, 35695,
  140151, 568503, 2390479, 10349535, 46206735,
  211799311, 997313823, 4809701439, 23758664095}
```

Consider *reversion*, $f(z) = 1/z$, an involution that maps a nonzero complex number to its reciprocal. Let U be the unit circle given by the equation $|z| = 1$ or $x^2 + y^2 = 1$, with $z = x + yi$. It is easy to see that f maps the interior of U to its exterior and vice versa; that is, $|z| > 1$ iff $|f(z)| < 1$. Moreover, $|f(z)| = 1$ iff $|z| = 1$. Reversion maps the number $z = x + yi$ into the number $\bar{z}/|z|^2$, where \bar{z} is the conjugate of z . Considering a complex number as a point in the complex plane, if a point is close to the origin, its reversion is far away, and vice versa.

Here is the reversion of a complex number.

```
ComplexExpand[ $\frac{1}{x + y \, i}$ ]
```

$$\frac{x}{x^2 + y^2} - \frac{i y}{x^2 + y^2}$$


Inversion, $f(z) = 1/\bar{z}$, is an involution similar to reversion. For inversion, 0 , z and $f(z)$ are collinear for nonzero z . The point $z = x + yi$ inverts to the point $z/|z|^2$.

$$\text{ComplexExpand}\left[\text{Conjugate}\left[\frac{1}{x + y i}\right]\right]$$

$$\frac{x}{x^2 + y^2} + \frac{i y}{x^2 + y^2}$$

The following `Manipulate` compares reversion, inversion, and some similar transformations. Drag the point z ; the arrow starts at z and ends at $f(z)$ for the chosen transformation f ; the interior of U is colored blue.

```
Manipulate[
Module[
{x, y, r, q},
If[Chop[Norm[Z]] == 0, Z = {0, .001}];
{x, y} = Z;
r = Norm[Z];
q = Switch[T, 1, {x, -y}/r^2, 2, {x, y}/r^2, 3,
{x, y}/r, 4, {x^2 - y^2, -2 x y}/r^4, 5, {x^2 - y^2, 2 x y}/r^4,
6, {Cos[1/2 ArcTan[x, y]], -Sin[1/2 ArcTan[x, y]]}/Sqrt[r];
Graphics[{EdgeForm[Thick], ColorData[2, 6], Disk[],
Blue, Arrow[{Z, q}], Red, Disk[Z, .05], Disk[q, .05],
Style[{Text[TraditionalForm[z], Z, {0, -1.5}],
Text[TraditionalForm[f[z]], q, {-2, 0}]}], Black,
14}],
Axes -> True, PlotRange -> 3]
],
{{Z, {.5, .3}}, Locator, Appearance -> None},
Row[
{Control@{{T, 1, "f(z) = "},
{1 -> 1/z, 2 -> 1/Z, 3 -> Z/|Z|, 4 -> 1/Z^2, 5 -> 1/Z, 6 -> 1/Sqrt[Z]},
SetterBar}, Spacer[20], "Drag the point z."}
],
TrackedSymbols -> {T, Z}, SaveDefinitions -> True]
```



■ Generalized Circles and Ellipses

A linear transformation maps the family of lines to itself. The following theorem suggests a family that may be preserved by inversion.

Theorem 1

Points (x, y) that satisfy the equation $A(x^2 + y^2) + Bx + Cy + D = 0$, with A, B, C, D real numbers, invert into points that satisfy the equation $D(x^2 + y^2) + Bx + Cy + A = 0$.

Applying the transformation corresponding to inversion of Cartesian points verifies this.

$$\text{Simplify}\left[\frac{A(x^2 + y^2) + Bx + Cy + D}{x^2 + y^2}, \left\{x \rightarrow \frac{x}{x^2 + y^2}, y \rightarrow \frac{y}{x^2 + y^2}\right\}\right]$$

$$\frac{A + Bx + D x^2 + Cy + D y^2}{x^2 + y^2}$$

Let $\Theta(A, B, C, D)$ be the set of points that satisfy the first equation in theorem 1. This four-parameter family includes points (for instance, when $A = 1$ and $D = B^2 + C^2$), lines (when $A = 0$), circles (when $A = 1$), the whole plane (when $A = B = C = D = 0$), and the empty set (when $A = B = C = 0$, $D = 1$).

Define a *generalized circle* to be a line or circle. (The function `genCircle` defined below constructs the graphics objects to draw the line or circle.) Under inversion, the generalized circle $\Theta(A, B, C, D)$ transforms into the generalized circle $\Theta(D, B, C, A)$. If $A = 0$, (and A, B are not simultaneously 0), then $\Theta(A, B, C, D)$ is a line. If $A \neq 0$, the equation $A(x^2 + y^2) + Bx + Cy + D = 0$ corresponds to a circle with center $-(B, C)/(2A)$ and radius $\sqrt{B^2 + C^2 - 4AD} / |2A|$, provided that $B^2 + C^2 \geq 4AD$. If $D = 0$, $\Theta(A, B, C, D)$ passes through the origin. For any nonzero k , $\Theta(A, B, C, D)$ is the same as $\Theta(kA, kB, kC, kD)$.

Other families of polynomials in two variables that include generalized circles do not transform their members into members of the same family; that is, they are not closed under inversion. For instance, the six-parameter family of equations $Ax^2 + By^2 + Cxy + Dx + Ey + G = 0$ includes conic sections, but some invert into quartics.

$$\begin{aligned} & \text{Simplify}\left[\frac{Ax^2 + By^2 + Cxy + Dx + Ey + G}{(x^2 + y^2)^2}, \{x \rightarrow x/(x^2 + y^2), y \rightarrow y/(x^2 + y^2)\}\right] \\ & \frac{1}{(x^2 + y^2)^2} \\ & (Ax^2 + Gx^4 + Cxy + Fx^2y + By^2 + 2Gx^2y^2 + Fy^3 + Gy^4 + Dx(x^2 + y^2)) \end{aligned}$$

Theorem 2

A line passing through different points (x_1, y_1) and (x_2, y_2) corresponds to the generalized circle $\Theta(0, y_2 - y_1, x_1 - x_2, x_2y_1 - x_1y_2)$ if $y_1 \neq y_2$ and $\Theta(0, 0, 1, -y_1)$ otherwise.

The parameters needed to describe such a line are obtained by solving the following equation.

$$\text{Solve}[x_1 + cy_1 + d == x_2 + cy_2 + d == 0, \{c, d\}]$$

$$\left\{ \left\{ c \rightarrow -\frac{x_1 - x_2}{y_1 - y_2}, d \rightarrow -\frac{x_2y_1 - x_1y_2}{y_1 - y_2} \right\} \right\}$$

Theorem 3

A circle with center (α, β) and radius ρ corresponds to the generalized circle $\Theta(1, -2\alpha, -2\beta, \alpha^2 + \beta^2 - \rho^2)$.

This result is readily obtained by expanding the Cartesian equation of such a circle.

$$\mathbf{Expand}[(\mathbf{x} - \alpha)^2 + (\mathbf{y} - \beta)^2 == \rho^2]$$

$$\mathbf{x}^2 + \mathbf{y}^2 - 2\mathbf{x}\alpha + \alpha^2 - 2\mathbf{y}\beta + \beta^2 == \rho^2$$

From a structural point of view, if a generalized circle is mapped to itself, it is said to be *preserved* by inversion. (Warning: that is not the same as saying that a point on a generalized circle is mapped to itself; for instance, a rotation about the center of a circle preserves the circle but moves each point of the circle.) In that case, $A = D$. If $A = D = 0$, the line passes through the origin and each point is inverted into its negative. If $A = D \neq 0$, and $B^2 + C^2 \geq 4A^2$, the circle $A(x^2 + y^2) + Bx + Cy + A = 0$ is preserved. Assume its center is at a point of the form $(h, 0)$. Then, in particular, the point $(h - \rho, 0)$ is inverted into the point $(h + \rho, 0)$.

$$\mathbf{Quiet@Solve}\left[\frac{\mathbf{d} - \rho}{(\mathbf{d} - \rho)^2} == \mathbf{d} + \rho, \{\mathbf{d}, \rho\}\right]$$

$$\left\{\left\{\rho \rightarrow -\sqrt{-1 + \mathbf{d}^2}\right\}, \left\{\rho \rightarrow \sqrt{-1 + \mathbf{d}^2}\right\}\right\}$$

Hence $\rho^2 + 1 = d^2$; that is, the origin, center, and either of the two intersection points of the circle and the unit circle U form a right triangle; hence the circle is orthogonal to U . This condition turns out to be necessary and sufficient for a generalized circle to be preserved. Theorem 1 implies the following results, which also apply conversely.

1. Any circle not passing through the origin inverts into a circle not passing through the origin (in fact passing through its intersection points with the unit circle U , if any).
2. Any circle passing through the origin inverts into a line not passing through the origin (in fact passing through its intersection points with U , if any; in general, this line is parallel to the tangent of the circle at the origin).
3. Any line passing through the origin is preserved by inversion (any point is mapped into its negative).
4. Any line not passing through the origin inverts into a circle passing through the origin (and its intersection points with U , if any).

To make inversion continuous, define the inversion of the origin to be a legal point, the “point at infinite,” ∞ , making the phrase “nonzero” unnecessary when talking about inversion. Inversion is thus a one-to-one map of the extended plane.

Conversions of generalized circles to and from graphics primitives are handled by the following functions.

```

genCircle[{}] = {};

genCircle[{0, B_, C_, D_}] :=
  If[Chop[B] == 0, If[Chop[C] == 0, {},
    Line[{{0, -D / C}, {1, -D / C}}]],
  If[Chop[C] == 0, Line[{{-D / B, 0}, {-D / B, 1}}],
    Line[{{0, -D / C}, {1, (D - B) / C}}]]];

genCircle[{A_, B_, C_, D_}] :=
  If[Chop[B2 + C2 - 4 A D] < 0, {},
    Circle[{-B, C} / (2 A),  $\sqrt{B^2 + C^2 - 4 A D} / \text{Abs}[2 A]$ ]];

toGenC[{}] = {};

toGenC[Circle[{α_, β_}, ρ_]] := {1, -2 α, -2 β, α2 + β2 - ρ2}

toGenC[Line[{{x1_, y1_}, {x2_, y2_}}]] :=
  If[(Chop[x1 - x2] == 0) ∧ (Chop[y1 - y2] == 0), {},
  If[Chop[y1 - y2] == 0, {0, 0, 1, -y1},
    If[Chop[x1 - x2] == 0, {0, 1, 0, -x1},
      {0, y2 - y1, x1 - x2, x2 y1 - x1 y2}]]];


invert[{}] = {};

invert[{A_, B_, C_, D_}] := {D, B, C, A}

```

For example, the following `Manipulate` shows how a line and a circle behave under inversion. To vary their positions, drag the two small disks. For a line, the disks are points on the line; for a circle, one disk is the center and the other is on the circumference. You can also choose a family of parallel lines or a family of concentric circles. The arrows show the action of the inversion on the two control points.

```
Manipulate[
Module[
{p, q, u, v, w, i, o1, o2},
{p, q} = pq;
If[Chop[Det[pq]] == 0, p = p + {0, .01};
q = q + {.01, 0}];
{u, v} = {p - 100 Normalize[q - p], p + 100 Normalize[q - p]};
w = {{0, -1}, {1, 0}}.Normalize[v - u];
o1 = Switch[op, 1, {Line[{u, v}]}, 2,
{Circle[p, Norm[p - q]]}, 3,
Table[Line[{u + i w, v + i w}], {i, -6, 6, .5}], 4,
Table[Circle[p, i Norm[p - q]], {i, .2, 6, .2}]];
o2 = Map[genCircle[invert[toGenC[#]]] &, o1];
Graphics[
{EdgeForm[Thick], ColorData[2, 6], Disk[], Red, o1,
Blue, o2, Disk[#, .05] & /@ {p, q}, Black,
If[arrows, {Arrow[{p, p / Norm[p]^2}],
Arrow[{q, q / Norm[q]^2}]}, {}]},
Axes -> (op < 3), PlotRange -> 3]
],
{{pq, {{2, 1}, {-2, 2}}}, Locator, Appearance -> None},
Row[{
Control@{{op, 1, ""}, {1 -> " line ", 2 -> " circle ",
3 -> " parallel lines ", 4 -> " concentric circles "}},
Spacer[20],
Control@{arrows, {True, False}}, Spacer[10],
"Drag a point."
}],
SaveDefinitions -> True]
```


Concentric circles do not invert into concentric circles, and the center of a circle does not invert into the center of its inversion (although the two centers are collinear with the origin). The center of the inverted circle can even be outside the original circle! (This particularly applies to U ; its interior gets mapped into its exterior and its center gets mapped to ∞ .) The next section shows how to locate the center of the inversion of a given circle.

As a circle is a particular kind of ellipse, it is interesting to see the inversion of a general ellipse. The following **Manipulate** shows such an inversion; you can drag the center of the ellipse with a locator and vary the lengths of its axes with a pair of sliders. Contour lines showing concentric ellipses are optional; increase the zoom slider if the display is all one color.

```

Manipulate[
Module[
{cx, cy},
{cx, cy} = c;
Show[Graphics[{EdgeForm[Thick], ColorData[2, 6],
Disk[{0, 0}, 1], Opacity[.5], ColorData[2, 5],
Disk[c, {a, b}]}], PlotRange → z],
If[c1,
If[c11 == 1, ContourPlot[ $\frac{(-cx + x)^2}{a^2} + \frac{(-cy + y)^2}{b^2}$ ,
{x, -8, 8}, {y, -8, 8}, Contours → 10],
ContourPlot[ $\frac{\left(-cx + \frac{x}{x^2+y^2}\right)^2}{a^2} + \frac{\left(-cy + \frac{y}{x^2+y^2}\right)^2}{b^2}$ , {x, -8, 8},
{y, -8, 8}]
],
ContourPlot[ $\frac{\left(-cx + \frac{x}{x^2+y^2}\right)^2}{a^2} + \frac{\left(-cy + \frac{y}{x^2+y^2}\right)^2}{b^2} == 1$ ,
{x, -8, 8}, {y, -8, 8}]
]
],
"Drag the locator.",
{{a, 1, "horizontal axis"}, .1, 4, Appearance → "Labeled"},
{{b, .6, "vertical axis"}, .1, 4, Appearance → "Labeled"},
{{c, {-.1, 0}}, Locator},
Dynamic@
Row[
Control[{{c1, False, "contour lines"}, {True, False}}],
If[c1,
Control[{{c11, 1, ""},
{1 → "family", 2 → "family inverted"}}], ""],
{{z, 4, "zoom"}, 2, 8, Appearance → "Labeled"},
SaveDefinitions → True]

```



■ Inversion in an Arbitrary Circle

Up to now, inversion has been in the unit circle; it can be generalized to use any circle M . Temporarily using complex numbers to represent points in the plane, assume $M = \odot(\gamma, \rho)$, which means M has center γ and radius ρ (assume $\rho > 0$). The generalization takes three steps: first transform M into U by translating γ to the origin and scaling by $1/\rho$. Second, invert in U . Third, scale back by ρ and translate back by γ . As these operations are rigid motions, all the properties of inversion apply to the generalized inversion. Generalized inversion is implemented by the function `invZ`.

$$\text{invZ}[\{\gamma_, \rho_ \}, z_] := \gamma + \text{Conjugate}\left[\frac{\rho^2}{z - \gamma}\right]$$

The explicit Cartesian coordinates for inverting points come from expanding the following expression.

ComplexExpand[**invZ**[{ $\gamma x + \gamma y \mathbf{i}$, ρ }, $x + y \mathbf{i}$]]

$$\gamma x + \frac{x \rho^2}{(x - \gamma x)^2 + (-y + \gamma y)^2} - \frac{\gamma x \rho^2}{(x - \gamma x)^2 + (-y + \gamma y)^2} + \\ \mathbf{i} \left(\gamma y + \frac{y \rho^2}{(x - \gamma x)^2 + (-y + \gamma y)^2} - \frac{\gamma y \rho^2}{(x - \gamma x)^2 + (-y + \gamma y)^2} \right)$$

Then, operating with Cartesian coordinates, the function **inver** inverts a given point p in $M = \odot(\gamma, \rho)$.

inver[{ γ _, _}, γ _] = {};

inver[{ γ _, ρ _}, p _] := $\gamma + \frac{\rho^2 (p - \gamma)}{(p - \gamma) \cdot (p - \gamma)}$

For example, the functions give matching results when applied to corresponding representations of the same point in the plane. The latter form is more convenient here.

invZ[{ $1 + 2 \mathbf{i}$, 3 }, $4 + 5 \mathbf{i}$]

$$\frac{5}{2} + \frac{7 \mathbf{i}}{2}$$

inver[{{ $1, 2$ }, 3 }, { $4, 5$ }]

$$\left\{ \frac{5}{2}, \frac{7}{2} \right\}$$

An interesting property of inversion is adopted by many authors as an alternative way of defining inversion. Let q be the inversion of the point $p \neq \gamma$ in the circle $M = \odot(\gamma, \rho)$. It is easy to verify that the product of the distance of p to γ and the distance of q to γ is a constant equal to ρ^2 .

rule1 = { $p \rightarrow \{px, py\}$, $\gamma \rightarrow \{\gamma x, \gamma y\}$ };

Simplify[**Norm**[γ - **inver**[{ γ , ρ }, p]] **Norm**[γ - p] /. **rule1**] /.

$$\text{Abs}[\mathbf{xx_}]^2 \rightarrow \mathbf{xx}^2, \left((\mathbf{px} - \gamma \mathbf{x})^2 + (\mathbf{py} - \gamma \mathbf{y})^2 > 0 \right) \bigwedge (\rho > 0) \Big]$$

ρ^2

Therefore, define the point q as the inverse of p if q is that unique point such that its distance to γ times the distance of p to γ is ρ^2 [7].

Before obtaining the Cartesian coordinates corresponding to inverting circles and lines, first consider the following problem. What is the center of the circle through three different noncollinear points a, b, c ?

The function `disSq` computes the square of the distance between two given points h, k and is used by the function `cir3` that computes the required center.

```
disSq[h_, k_] := (k - h) . (k - h)

cir3[a_, b_, c_] := Module[{xx, yy},
  First[
    {xx, yy} /.
    Solve[disSq[{xx, yy}, a] == disSq[{xx, yy}, b] ==
      disSq[{xx, yy}, c], {xx, yy}]]]
```

For example, here is an explicit formula for the center of such a circle through $a = (x_1, y_1)$, $b = (x_2, y_2)$, and $c = (x_3, y_3)$.

```
simplifiedcenter =
Simplify[cir3[{x1, y1}, {x2, y2}, {x3, y3}] /.
  Abs[xx_]² → xx²]
```

$$\left\{ \left(x_3^2 (-y_1 + y_2) + x_2^2 (y_1 - y_3) - \left(x_1^2 + (y_1 - y_2)(y_1 - y_3) \right) (y_2 - y_3) \right) / \right. \\ \left. \left(2 (x_3 (-y_1 + y_2) + x_2 (y_1 - y_3) + x_1 (-y_2 + y_3)) \right), \right. \\ \left. \left(-x_2^2 x_3 + x_1^2 (-x_2 + x_3) + x_3 (y_1^2 - y_2^2) + \right. \right. \\ \left. \left. x_1 (x_2^2 - x_3^2 + y_2^2 - y_3^2) + x_2 (x_3^2 - y_1^2 + y_3^2) \right) / \right. \\ \left. \left(2 (x_3 (y_1 - y_2) + x_1 (y_2 - y_3) + x_2 (-y_1 + y_3)) \right) \right\}$$

This is its radius.

```
FullSimplify[Norm[simplifiedcenter - {x1, y1}] /.
  Abs[xx_]² → xx²]
```

$$\frac{1}{2} \sqrt{\left(\left((x_1 - x_2)^2 + (y_1 - y_2)^2 \right) \right. \\ \left. \left((x_1 - x_3)^2 + (y_1 - y_3)^2 \right) \left((x_2 - x_3)^2 + (y_2 - y_3)^2 \right) \right) / \\ \left(x_3 (-y_1 + y_2) + x_2 (y_1 - y_3) + x_1 (-y_2 + y_3) \right)^2}$$

The denominators of the expressions for the center and radius contain a factor that is zero only when a, b , and c are collinear.

```
rule2 = {a → {x1, y1}, b → {x2, y2}, c → {x3, y3}};
Det[{c - a, b - a} /. rule2]
```

$$x_2 y_1 - x_3 y_1 - x_1 y_2 + x_3 y_2 + x_1 y_3 - x_2 y_3$$

Now consider an arbitrary circle V that does not pass through the center γ of M . Invert γ in V to γ' and then invert γ' in M to γ'' . Then the inversion of V in M is a circle with center γ'' . The equality of the following two results verifies this property.

```
rule3 = {γ → {γx, γy}, M → {{γx, γy}, ρ}, V → {{vx, vy}, vr}};
FullSimplify[inver[M, inver[V, γ]] /. rule3]
```

$$\left\{ \gamma x + \frac{(vx - \gamma x) \rho^2}{-vr^2 + (vx - \gamma x)^2 + (vy - \gamma y)^2}, \right. \\ \left. \gamma y + \frac{(vy - \gamma y) \rho^2}{-vr^2 + (vx - \gamma x)^2 + (vy - \gamma y)^2} \right\}$$

```
rule4 = {M → {{γx, γy}, ρ}};
FullSimplify[cir3[inver[M, {vx + vr, vy}] /. rule4,
  inver[M, {vx - vr, vy}] /. rule4,
  inver[M, {vx, vy + vr}] /. rule4]]
```

$$\left\{ \gamma x + \frac{(vx - \gamma x) \rho^2}{-vr^2 + (vx - \gamma x)^2 + (vy - \gamma y)^2}, \right. \\ \left. \gamma y + \frac{(vy - \gamma y) \rho^2}{-vr^2 + (vx - \gamma x)^2 + (vy - \gamma y)^2} \right\}$$

Theorem 4

If the circle $V = \odot(c, \sigma)$ inverts in $M = \odot(\gamma, \rho)$ into the circle W , then the center of W is $\gamma + (c - \gamma)\rho^2 / (|c - \gamma|^2 - \sigma^2)$ and its radius is $\sigma\rho^2 / |c - \gamma|^2 - \sigma^2|$.

The vertical bars in the last expression have two different meanings: the modulus of a vector and the absolute value of a number. Theorem 4 arises from using the property mentioned in the previous paragraph. First we obtain the square of the radius of W .

```
rule5 = {γ → {γx, γy}, M → {{γx, γy}, ρ}, c → {cx, cy},
  V → {{cx, cy}, σ}};
```

```
FullSimplify[
  disSq[inver[M, inver[V, γ]], inver[M, {cx + σ, cy}]] /.
  rule5 /. Abs[xx_]^2 → xx^2]
```

$$\frac{\rho^4 \sigma^2}{((cx - \gamma x)^2 + (cy - \gamma y)^2 - \sigma^2)^2}$$

Match the center of W with c to get the following.


```
FullSimplify@Reduce[(c == inver[M, inver[V, γ]]) /. rule5]
```

$$\left(\left(cx + \sqrt{-(cy - \gamma y)^2 + \rho^2 + \sigma^2} == \gamma x \mid \mid \right. \right. \\ \left. \left. \gamma x + \sqrt{-(cy - \gamma y)^2 + \rho^2 + \sigma^2} == cx \right) \&\& \rho \neq 0 \right) \mid \mid \\ (cy == \gamma y \&\& cx == \gamma x \&\& \sigma \neq 0)$$

So a circle $V = \odot(c, \sigma)$ is inverted into a concentric circle if c is outside the circle of inversion $M = \odot(\gamma, \rho)$ and $|c - \gamma| = \sigma^2 + \gamma^2$. This implies that M and V are orthogonal. Therefore, a circle inverts into a concentric circle (and necessarily of the same radius, i.e. itself) iff it is orthogonal to M . To visualize these ideas, consider the following Manipulate that inverts a pattern of tangent circles.

```
invCir[M: {γ_, ρ_}, {cV_, rV_}] := Module[
  {n = disSq[cV, γ] - rV^2, h, k},
  If[n == 0,
    h = inver[M, cV + rV Normalize[cV - γ]];
    k = {{0, -1}, {1, 0}}.Normalize[cV - γ];
    Line[{h + 100 k, h - 100 k}],
    Circle[γ + (cV - γ) ρ^2 / n, rV ρ^2 / Abs[n]]
  ]
]

Manipulate[
  Module[
    {g, i, j},
    g =
      Join[Flatten[Table[{{i, j} - (n + 1) / 2, .5}, {i, n},
        {j, n}], 1],
        Flatten[Table[{{i, j} - (n + 1) / 2 + .5 {1, 1}, .2071},
          {i, n - 1}, {j, n - 1}], 1]];
    Graphics[{EdgeForm[Thick], Opacity[.5], ColorData[2, 6],
      Disk[γ, ρ], ColorData[5, 7],
      If[io, Map[(invCir[{γ, ρ}, #] /. Circle → Disk) &, g], {}],
      Black, If[go, Circle@@# & /@g, {}]], PlotRange → z,
      ImagePadding → 1]
  ],
  Row[{"Drag the locator.", Spacer[50],
    Control@{{n, 4, "complexity"}, Range[10], Setter}},
  {{ρ, .5, "radius of circle of inversion"}, 0, 5, Slider},
  Row[{
    Control[{{go, True, "grid"}, {True, False}}],
    Spacer[3],
    Control[{{io, True, "inversion"}, {True, False}}],
    Spacer[4], Control[{{z, 2, "zoom"}, .01, 6, Slider}]
  ]},
  {{γ, {0, 0}}, Locator},
  SaveDefinitions → True]
```



■ Inverting Segments and Arcs

The inversion of a polygon is a closed sequence of arcs of circles that go through the center of inversion. Consider the problem of finding an arc through three noncollinear points a , b , and c . As stated, this problem has many solutions; by assuming one of the points is not an endpoint of the arc, there is a unique answer. The function `seg3` constructs such an arc and includes an argument to control whether the arc passes through the point b .

```
cir3N[a_, b_, c_] := Module[{xx, yy},
  First[
    {xx, yy} /.
      NSolve[disSq[{xx, yy}, a] == disSq[{xx, yy}, b] ==
        disSq[{xx, yy}, c], {xx, yy}]
  ]

  at[{x_, y_}] := Module[{u = ArcTan[x, y]}, u + If[u < 0, 2  $\pi$ , 0]]
```



```

seg3[a_, b_, c_, over_] := Module[{ce, aa, ac},
  ce = cir3N[a, b, c];
  {aa, ac} = at /@ {a - ce, c - ce};
  If[aa > ac, ac = ac + 2  $\pi$ ];
  If[Det[{b - a, c - a}] < 0, {aa, ac} = {ac, aa};
    If[aa > ac, aa = aa - 2  $\pi$ ]];
  If[! over, {aa, ac} = {ac, aa + 2  $\pi$ ]];
  Circle[ce, Norm[ce - a], {aa, ac}]
]


```

The following Manipulate applies the function seg3 to any three noncollinear points, optionally passing through one of them.

```

Manipulate[
  Module[{a, b, c, ce},
    {a, b, c} = abc;
    If[Chop[Det[{b - a, c - a}]] == 0,
      {a, b, c} = abc = abc + RandomReal[{- .01, .01}, {3, 2}]];
    ce = cir3N[a, b, c];
    Graphics[{Thick, {Dotted, Circle[ce, Norm[ce - a]]},
      seg3[a, b, c, ox == 1], Red,
      EdgeForm[Thin], Disk[a, .03], Disk[c, .03],
      Text[Style["x", Bold, 30], b]}, PlotRange -> 1]
  ],
  {{abc, {{-.2, -.8}, {.5, -.2}, {-.1, .4}}}, Locator,
  Appearance -> None},
  Row[
    {Control[{{ox, 1, "passing over x?"},
      {1 -> " yes ", 2 -> " no "}}, Spacer[30],
      "Drag a red point."}},
    SaveDefinitions -> True]

```



Let the inversion of the point h be h' and the point k be k' . There are four cases for the inversion of the finite line segment hk in the circle $M = \odot(\gamma, \rho)$.

1. If $h = k$, the segment is a point and inverts into the point $h' = k'$.
2. If $h \neq k$ and $h = \gamma$, the segment hk inverts into a ray starting at k' , going away from q .
3. If h, k , and q are collinear, there are two subcases:
 - a. If q is in between h and k , the inverse is the union of two rays, one starting at h' , the other starting at k' , and both going away from γ .
 - b. Otherwise, h and k are on the same side of γ , and the inverse of hk is a line segment joining h' and k' on the same side of γ as h and k , but now switched around.
4. If h and k are not collinear with γ , consider the circle passing through γ , h' , and k' . The inversion of hk is the arc joining h' and k' that does not pass through γ .

Similar results apply to the inversion of an arc. The following `Manipulate` shows all these cases as they apply to line segments and arcs. The initial segments or arcs are shown in red and their inversions in blue. Control locators are drawn in red. When a segment/arc is such that its corresponding line/circle passes through γ , say that it does so *directly* if γ is part of the segment/arc and *indirectly* otherwise.

```

di[p_, col_] := {If[col == 1, Red, ColorData[2, 7]],
  Disk[p, .08]}

li[s_, col_] := {If[col == 1, Red, ColorData[2, 7]], Line[s]}


Manipulate[
Module[{M = {{0, 0}, 1}, h, k, i, u, v},
{h, k, i} = hki;
(* h, k, i and {0, 0} must be all different *)
If[Length[Union[Chop/@{h, k, i, {0, 0}}]] < 4,
hki = {h, k, i} = {h, k, i} + RandomReal[{- .01, .01}, {3, 2}]
];
{u, v} = {inver[M, h], inver[M, k]};
Graphics[{EdgeForm[Black], ColorData[2, 6], Disk[],
Thick, Black,
Switch[cases,
1, {Style[Text["both ends equal", {0, -2.8}, {0, 0}],
15], Arrow[{h, u}], di[h, 1], di[u, 2]},
2, {
Style[Text["an end on the center of inversion",
{0, -2.8}, {0, 0}], 15],
If[
segArcs == 1,
{li[{h, {0, 0}}, 1], li[{u, u + 100 Normalize[u]}, 2],
di[u, 2]},
{Red, seg3[{0, 0}, h, k, True], di[k, 1],
li[{u + 100 Normalize[u - v], v}, 2]}
],
di[h, 1]
},
3, {
Style[
Text["directly covering the center of inversion",
{0, -2.7}, {0, 0}], 15],
If[
segArcs == 1,
{
k = -Abs[k.Normalize[h]] Normalize[h];
v = inver[M, k];
li[{h, k}, 1],
li[{u, u + 100 Normalize[u]}, 2],
li[{v, v + 100 Normalize[v]}, 2]
},
{
Red, seg3[h, {0, 0}, k, True],
li[{u, u + 100 Normalize[u - v]}, 2],
}
}
}
]

```

```

        li[{v, 100 Normalize[v - u]}, 2]
    }
],
di[h, 1], di[k, 1], di[u, 2], di[v, 2]
},
4, {
Style[
Text["indirectly covering the center of inversion",
{0, -2.7}, {0, 0}], 15],
If[
segArcs == 1,
{
k = Abs[k.Normalize[h]] Normalize[h];
v = inver[M, k];
li[{h, k}, 1],
li[{u, v}, 2]
},
{Red, seg3[h, {0, 0}, k, False], li[{u, v}, 2]}
],
di[h, 1], di[k, 1], di[u, 2], di[v, 2]
},
5, {
Style[
Text["not passing through the center of inversion",
{0, -2.7}, {0, 0}], 15],
If[
segArcs == 1,
{li[{h, k}, 1], seg3[u, inver[M, (h + k) / 2],
v, True]},
{
Red, seg3[h, i, k, False],
ColorData[2, 7], seg3[u, inver[M, i], v, False],
di[i, 1]], di[h, 1], di[k, 1], di[u, 2], di[v, 2]
}
],
Brown, Disk[{0, 0}, .05]], PlotRange → 3]
],
{{segArcs, 1, Spacer[30]},
{1 → " line segments ", 2 → " arcs "}},
Row[
Control@{{cases, 1}, {1 → " 1 ", 2 → " 2 ", 3 → " 3a ",
4 → " 3b ", 5 → " 4 "}},
Spacer[30], "Drag a red point."],
{{hki, {{-1.5, -1}, {-1, -1}, {2, 2}}}, Locator,
Appearance → None},
SaveDefinitions → True]

```



■ Inverting a Triangle and a Quadrilateral

The function `invSeg` encapsulates the four cases appearing in the inversion of a segment mentioned in the previous section.

```
invSeg[M_, {h_, h_}] := inver[M, h]
```

```
invSeg[M: {γ_, _}, {γ_, k_}] :=  
  Module[{u = inver[M, k]}, Line[{u, u + 100 Normalize[k - γ]}]]
```

```
invSeg[M: {γ_, _}, {h_, γ_}] := invSeg[M, {γ, h}]
```

```

invSeg[M: {γ_, _}, {h_, k_}] :=
Module[{h1, k1}, (* h,k,q collinear *)
{h1, k1} = {inver[M, h], inver[M, k]};
If[
Norm[h - γ] + Norm[k - γ] == Norm[h - k],
(* q is between h and k *)
{Line[{h1, h1 + 100 Normalize[h - γ]}],
Line[{k1, k1 + 100 Normalize[k - γ]}]},
Line[{h1, k1}]
]
] /; Det[{h - γ, k - γ}] == 0

invSeg[M_, {h_, k_}] :=
Module[{a, c, ce, aa, ac}, (* h,k,γ not collinear *)
{h1, k1} = {inver[M, h], inver[M, k]};
seg3[h1, inver[M, (h + k) / 2], k1, True]
]

invAll[M_, p_] :=
Quiet@Map[invSeg[M, #] &, Partition[p, 2, 1, 1]]

```

The function `invSeg` is used to invert a general triangle and a general quadrilateral. As mentioned before, the inversion of a polygon is a figure made by adjoining arcs of circles that pass through q . (The geometry of coincident arcs is the same as the geometry of polygons. Just invert them!) However, the interior of a polygon does not invert to the interior of its inverse. The function `fill` fills this interior for aesthetics.

```

fill[{
Circle[c1_, r1_, {α1_, β1_}],
Circle[c2_, r2_, {α2_, β2_}],
Circle[c3_, r3_, {α3_, β3_}]
}] :=
Module[{t, t1, t2, t3, t4, mi, d1, d2, d3, d4},
t1 = Table[c1 + r1 {Cos[t], Sin[t]},
{t, α1, β1, .01 (β1 - α1)}];
t2 = Table[c2 + r2 {Cos[t], Sin[t]},
{t, α2, β2, .01 (β2 - α2)}];
t3 = Table[c3 + r3 {Cos[t], Sin[t]},
{t, α3, β3, .01 (β3 - α3)}];
{d1, d2, d3, d4} =
Norm /@ {Last[t1] - First[t2], Last[t1] - Last[t2],
Last[t1] - First[t3], Last[t1] - Last[t3]};
mi = Min[d1, d2, d3, d4];
t4 = Join[t1, Which[mi == d1, t2, mi == d2, Reverse[t2],
mi == d3, t3, True, Reverse[t3]]];
If[(mi == d3) ∨ (mi == d4), t3 = t2];
{d1, d2} =
Norm /@ {Last[t4] - First[t3], Last[t4] - Last[t3]};
mi = Min[d1, d2];
Polygon[Join[t4, Which[mi == d1, t3, True, Reverse[t3]]]]
]

```


```

fill[
{
  Circle[c1_, r1_, {α1_, β1_}],
  Circle[c2_, r2_, {α2_, β2_}],
  Circle[c3_, r3_, {α3_, β3_}],
  Circle[c4_, r4_, {α4_, β4_}]]] :=
Module[{t, t1, t2, t3, t4, t5, t6, mi, d1, d2, d3, d4,
  d5, d6},
  t1 = Table[c1 + r1 {Cos[t], Sin[t]},
    {t, α1, β1, .01 (β1 - α1)}];
  t2 = Table[c2 + r2 {Cos[t], Sin[t]},
    {t, α2, β2, .01 (β2 - α2)}];
  t3 = Table[c3 + r3 {Cos[t], Sin[t]},
    {t, α3, β3, .01 (β3 - α3)}];
  t4 = Table[c4 + r4 {Cos[t], Sin[t]},
    {t, α4, β4, .01 (β4 - α4)}];
  {d1, d2, d3, d4, d5, d6} =
    Norm /@ {Last[t1] - First[t2], Last[t1] - Last[t2],
      Last[t1] - First[t3], Last[t1] - Last[t3],
      Last[t1] - First[t4], Last[t1] - Last[t4]};
  mi = Min[d1, d2, d3, d4, d5, d6];
  t5 = Join[t1,
    Which[
      mi == d1, t2,
      mi == d2, Reverse[t2],
      mi == d3, t3,
      mi == d4, Reverse[t3],
      mi == d5, t4,
      True, Reverse[t4]
    ]
  ];
  If[(mi == d3) ∨ (mi == d4), t3 = t2];
  If[(mi == d5) ∨ (mi == d6), t4 = t2];
  {d1, d2, d3, d4} =
    Norm /@ {Last[t5] - First[t3], Last[t5] - Last[t3],
      Last[t5] - First[t4], Last[t5] - Last[t4]};
  mi = Min[d1, d2, d3, d4];
  t6 = Join[t5, Which[mi == d1, t3, mi == d2, Reverse[t3],
    mi == d3, t4, True, Reverse[t4]]];
  If[(mi == d3) ∨ (mi == d4), t4 = t3];
  {d1, d2} =
    Norm /@ {Last[t6] - First[t4], Last[t6] - Last[t4]};
  mi = Min[d1, d2];
  Polygon[Join[t6, Which[mi == d1, t4, True, Reverse[t4]]]]
]

```

The following `Manipulate` shows a triangle or quadrilateral and the corresponding filled inversion. You can zoom by changing the radius of the circle of inversion M . For what positions of q do the sides of a given triangle invert into three congruent circles? Hint: it has something to do with its incircle. What about a quadrilateral?

```
Manipulate[
Module[{v, c, iA},
(* all locators must be different *)
If[Length[Union[Chop/@cpqrs]] < 5,
cpqrs = cpqrs + RandomReal[{- .01, .01}, {5, 2}]];
v = Take[Rest[cpqrs], If[op == 1, 3, 4]];
c = First[cpqrs];
iA = invAll[{c, r}, v];
Graphics[{EdgeForm[Thin], Opacity[.5], ColorData[2, 6],
Disk[c, r], ColorData[4, 3], Polygon[v], Opacity[1],
Red, Disk[#, .07] & /@ v, Disk[c, .07], Black,
Map[Arrow[#, inver[{c, r}, #]] &, v], Opacity[.7],
Yellow, fill[iA], Black, Thick, iA,
{Red, Red, Disk[#, .07] & /@ v, Disk[c, .07]}},
PlotRange -> 3]
],
{{cpqrs, 1. {{0, 0}, {-1.5, 2}, {-1, -2.5}, {2, -1},
{2.5, 0}}}, Locator, Appearance -> None},
{{r, 1, "radius of inversive circle"}, .001, 3,
Appearance -> "Labeled"},
Row[{Spacer[140], "Drag a red point."}],
{{op, 1, ""}, {1 -> " triangle ", 2 -> " quadrilateral "}},
SaveDefinitions -> True]
```

■ Inverting the Visual Proof of Nicomachus's Theorem


Nicomachus's theorem [8, 9] states that $\sum_{k=1}^n k^3 = (\sum_{k=1}^n k)^2$. This section inverts a well-known pattern showing a proof without words of this identity.

To justify the identity, consider the pattern made by half-squares and squares of increasing integer side lengths. The area of the big square is the square of the sum of the side lengths on the base $1 + 2 + 3 + \dots + 7$. On the other hand, this area is the sum of the individual areas of the interior squares, taking into account that two half-squares make a whole square: $1 + 2 \times 2^2 + 3 \times 3^2 + \dots + 7 \times 7^2$. Logically these quantities are the same; hence Nicomachus's theorem.

```


Module[{b, n = 7},
Graphics[{EdgeForm[Thin], White,
  Table[
    b = i (i - 1) / 2;
    {
      Rectangle[{b, 0}, {b + i, i / 2}],
      Table[Rectangle[{b, i / 2 + i j}, {b + i, i / 2 + i (j + 1)}],
        {j, 0, i / 2 - 1}],
      Rectangle[{0, b}, {i / 2, b + i}],
      Table[Rectangle[{i / 2 + i j, b}, {i / 2 + i (j + 1), b + i}],
        {j, 0, i / 2 - 2}]
    }, {i, 2, n, 2}
  ],
  Table[b = i (i - 1) / 2;
    {
      Table[Rectangle[{b, i j}, {b + i, i j + i}], {j, 0, i / 2}],
      Table[Rectangle[{i j, b}, {i j + i, b + i}],
        {j, 0, i / 2 - 1}]], {i, 1, n, 2}
    ]
  ]]
]

```



In order to get a symmetrical version, promote the half-squares into squares.

```
Module[{b, n = 7},
Graphics[{EdgeForm[Thin], White,
Table[b = i (i - 1) / 2;
{
Rectangle[{b, -i / 2}, {b + i, i / 2}],
Table[Rectangle[{b, i / 2 + i j}, {b + i, i / 2 + i (j + 1)}],
{j, 0, i / 2 - 1}],
Rectangle[{-i / 2, b}, {i / 2, b + i}],
Table[Rectangle[{i / 2 + i j, b}, {i / 2 + i (j + 1), b + i}],
{j, 0, i / 2 - 2}]
},
{i, 2, n, 2}],
Table[b = i (i - 1) / 2;
{
Table[Rectangle[{b, i j}, {b + i, i j + i}], {j, 0, i / 2}],
Table[Rectangle[{i j, b}, {i j + i, b + i}],
{j, 0, i / 2 - 1}]
}, {i, 1, n, 2}]
}]
]
```



Add replicas around the lower-left vertex of the unit square made by rotating this pattern 90, 180, and 270 degrees to obtain a symmetrical pattern. Then eliminate repeated squares; moreover, eliminate repeated segments from adjacent squares using `filter`.

```
filter = {
  {x___, {a : {ax_, ay_}, {bx_, ay_}}, y___,
   {{bx_, ay_}, c : {cx_, ay_}}, z___}  $\Rightarrow$  {x, {a, c}, y, z},
  {x___, {{bx_, ay_}, c : {cx_, ay_}}, y___,
   {a : {ax_, ay_}, {bx_, ay_}}, z___}  $\Rightarrow$  {x, {a, c}, y, z},
  {x___, {a : {ax_, ay_}, b : {bx_, ay_}}, y___,
   {{cx_, ay_}, {dx_, ay_}}, z___}  $\Rightarrow$ 
  {x, {a, b}, y, z} /; ax  $\leq$  cx  $\leq$  dx  $\leq$  bx
};
```

The function `nichoSegs` computes the segments needed to form the next pattern shown, which gives a visual proof of Nicomachus's theorem [9].

```
nichoSegs[n_] :=
Module[{m = {{0, -1}, {1, 0}}, g, h, a, b, ax, ay, bx,
  by, h2, xx, yy, u},
  g = {
    Table[b = i (i - 1) / 2;
      {
        Rectangle[{b, -i / 2}, {b + i, i / 2}],
        Table[Rectangle[{b, i / 2 + i j}, {b + i, i / 2 + i (j + 1)}],
          {j, 0, i / 2 - 1}],
        Rectangle[{i / 2, b}, {i / 2, b + i}],
        Table[Rectangle[{i / 2 + i j, b}, {i / 2 + i (j + 1), b + i}],
          {j, 0, i / 2 - 2}]
      },
    {i, 2, n, 2}],
  Table[b = i (i - 1) / 2;
    {
      Table[Rectangle[{b, i j}, {b + i, i j + i}],
        {j, 0, i / 2}],
      Table[Rectangle[{i j, b}, {i j + i, b + i}],
        {j, 0, i / 2 - 1}]
    },
    {i, 1, n, 2}]
  };
  h = Union[
    Flatten[{
      g,
      g /. Rectangle[a_, b_]  $\Rightarrow$  Rectangle[m.a, m.b],
      g /. Rectangle[a_, b_]  $\Rightarrow$  Rectangle[m.m.a, m.m.b],
      g /. Rectangle[a_, b_]  $\Rightarrow$  Rectangle[m.m.m.a, m.m.m.b]
    }] /. Rectangle[{ax_, ay_}, {bx_, by_}]  $\Rightarrow$ 
    Rectangle[{Min[ax, bx], Min[ay, by]},
      {Max[ax, bx], Max[ay, by]}]
  ];
  h2 = Union[Flatten[h /. Rectangle[{ax_, ay_}, {bx_, by_}]  $\Rightarrow$ 
    {
      {{ax, ay}, {bx, ay}},


```

```

      {{bx, ay}, {bx, by}},
      {{ax, by}, {bx, by}},
      {{ax, ay}, {ax, by}}}, 1]
];
u = Cases[h2, {{_, xx_}, {_, xx_}}] //. filter;
Join[u, u /. {xx_, yy_Integer} → {-yy, xx}]
]

Graphics[Line /@ nichoSegs[7]]

```



The function `nichoSegs` computes the minimum number of segments necessary to produce the pattern. This is the number of segments up to $n = 10$.

```


ParallelTable[Length[nichoSegs[n]], {n, 10}]

{6, 10, 18, 30, 46, 66, 90, 118, 150, 186}

```

The n^{th} pattern has $2(n^2 - n + 3)$ segments and $2n(n + 1)$ squares. Not removing unnecessary segments would give $8n(n + 1)$ segments, so for large n , removing unnecessary segments is more than four times better. The following `Manipulate` shows Nicomachus's pattern in gray along with its inversion in red. You can change M and thus q by dragging the locator, and you can vary r with the slider.

```
Manipulate[
Module[{nico, iA},
  nico = nichoSegs[n];
  If[io, iA = Map[invSeg[{c, r}, #] &, nico]];
  Graphics[{EdgeForm[Thick],
    If[ic, {Opacity[.5], ColorData[2, 6], Disk[c, r]}, {}],
    Gray, If[go, Line/@nico, {}], Red, Thick,
    If[io, iA, {}]], PlotRange -> z]
,
{{c, {0, 0}}, Locator},
Row[{"Drag the locator.", Spacer[40],
  Control@{{n, 2, "squares on the side"}, Range[7],
    Setter}}],
{{r, 1.3, "radius of circle of inversion"}, .01, 5,
  Slider, Appearance -> "Labeled"},
Row[{
  Control[{{go, True, "show grid"}, {True, False}}],
  Spacer[3],
  Control[{{io, True, "show its inversion"},
    {True, False}}], Spacer[3],
  Control[{{ic, True, "show circle of inversion"},
    {True, False}}]
}],
{{z, 2, "zoom"}, .01, 15, Slider, Appearance -> "Labeled"},
SaveDefinitions -> True]
```



■ Inversive Tiling with Equilateral Triangles, Squares, and Hexagons

There are only three ways to tile the plane with a regular polygon: using an equilateral triangle, a square, or a regular hexagon. The goal of this section is to show the patterns arising from inverting these tilings. To that end, the functions `ring3`, `ring4`, and `ring6` generate the corresponding set of pieces in a ring surrounding the circle of inversion (in red).

Here is the triangular case for the first level of complexity $n = 1$.


```

expand3[{x_, y_}] :=
  Module[{s =  $\sqrt{3}$ }, {{x + 2, y}, {x + 1, y + s}, {x - 1, y + s},
    {x - 2, y}, {x - 1, y - s}, {x + 1, y - s}}]

ring3[n_] :=
  Union@Flatten[NestList[Flatten[Join[expand3 /@ #], 1] &,
    {{0, 0}}, n], 1]

rule6 = {s →  $\sqrt{3}$ , M → {{1, 1 /  $\sqrt{3}$ }, 1 /  $\sqrt{3}$ }};


Graphics[{EdgeForm[Thin], Opacity[.5], Orange,
  Map[Polygon[{#, # + {2, 0}, # + {1,  $\sqrt{3}$ }}] &, ring3[1]],
  Red, Disk@@ (M /. rule6)}]
```



The corresponding functions `inv3`, `inv4`, and `inv6` invert each of the segments forming the tiling. For instance, here is the pattern for the triangular case.


```
inv3[M_, v_] := Module[{s =  $\sqrt{3}$ , m1 = M /. rule6},
  {
    invSeg[m1, {v, v + {2, 0}}],
    invSeg[m1, {v + {2, 0}, v + {1, s}}],
    invSeg[m1, {v + {1, s}, v}]
  }
]
```

```
Graphics[Map[inv3[M, #] &, ring3[1]]]
```




Although the interiors of triangles are not preserved by inversion, they are filled to show the interference patterns they produce. The idea is shown in the next figure; to avoid clutter, only the first four members of `ring3[1]` are drawn. Lines join the vertices of the triangles to their inversions.

```
With[{u = Take[ring3[1], 4], s = (s /. rule6), M = M /. rule6},
Graphics[{EdgeForm[Thin], Orange, Opacity[.5],
Map[Polygon[{#, # + {2, 0}, # + {1, s}}] &, u], Pink,
fill /@ Map[inv3[M, #] &, u], Black, Circle@@M,
Map[
{Line[{#, inver[M, #]}],
Line[{# + {2, 0}, inver[M, # + {2, 0}]}],
Line[{# + {1, s}, inver[M, # + {1, s}]]} &, u]}
]
]
```




Although the triangles do not overlap, the interiors of their inverses do. In fact, the inversion of the central triangle contains the interiors of all the rest. The next level of complexity renders the following pattern.

```
With[{M = (M /. rule6), u = ring3[2]},  
Graphics[{EdgeForm[Thin], ColorData[5, 6], Opacity[.5],  
fill /@ Map[inv3[M, #] &, u], Black, Circle@@M}  
]
```



Finally, a detail showing the sixth level.

```
With[{M = (M /. rule6), u = ring3[6]},
  Graphics[{EdgeForm[Thin], ColorData[5, 6], Opacity[.5],
    fill /@ Map[inv3[M, #] &, u]},
  PlotRange -> {{.725, 1.275}, {.357, .87}}]
]
```



Similarly, in the case of tiling with squares, here are the corresponding functions.

```



expand4[{x_, y_}] := {
  {x + 1, y},
  {x + 1, y + 1},
  {x, y + 1},
  {x - 1, y + 1},
  {x - 1, y},
  {x - 1, y - 1},
  {x, y - 1},
  {x + 1, y - 1}
}

ring4[n_] :=
  Union@Flatten[NestList[Flatten[Join[expand4 /@ #], 1] &,
    {{0, 0}}, n], 1]

inv4[M_, v_] := With[{m1 = (M /. rule6)},
  {
    invSeg[m1, {v, v + {1, 0}}],
    invSeg[m1, {v + {1, 0}, v + 1}],
    invSeg[m1, {v + 1, v + {0, 1}}],
    invSeg[m1, {v + {0, 1}, v}]
  }
]


With[{M = {{1, 1} / 2., 1 / 2.}, u = ring4[2]},
  GraphicsRow[{
    Graphics[{EdgeForm[Thin], Opacity [.5], Orange,
      Map[Polygon[{#, # + {1, 0}, # + 1, # + {0, 1}}] &, ring4[1]],
      Red, Disk@@M}],
    Graphics[{EdgeForm[Thin], ColorData[5, 6], Opacity [.5],
      fill /@ Map[inv4[M, #] &, u], Black, Circle@@M}]
  ]
]

```




Here is a detail corresponding to the first four squares of the arrangement `ring4[1]`.

```
With[{u = Take[ring4[1], 4], M = {{1, 1} / 2., 1 / 2.}},
Graphics[{EdgeForm[Thin], Orange, Opacity[.5],
Map[Polygon[#, # + {1, 0}, # + 1, # + {0, 1}] &, u],
Pink, fill /@ Map[inv4[M, #] &, u], Black, Circle@@M,
Map[{
Line[#, inver[M, #]],
Line[#, inver[M, # + {1, 0}]],
Line[#, inver[M, # + 1]],
Line[#, inver[M, # + {0, 1}]]
} &, u]
}]
]
```



And here is a detail of the sixth level. (The color assignment has to be made explicit and does not rely only on overlapping as it did before.)

```
Module[{M = {{1, 1} / 2., 1 / 2.}, u = ring4[6], i = 0},
Graphics[{
  EdgeForm[Thin], Opacity[.5],
  Map[
    {ColorData[2, If[EvenQ[i], 4, 5]], i++;
     fill[#]} &,
    Map[inv4[M, #] &, u]
  ]
}, PlotRange -> {{.25, .75}, {.25, .75}}]
```



The tiling using regular hexagons cannot be colored with two colors, and there are too many segments to place the center of M . So here is a line pattern.

```

expand6[{x_, y_}] := With[{s =  $\sqrt{3}$ },
  {
    {x + 3, y - s},
    {x + 3, y + s},
    {x, y + 2 s},
    {x - 3, y + s},
    {x - 3, y - s},
    {x, y - 2 s}
  }
]

ring6[n_] :=
  Union@Flatten[NestList[Flatten[Join[expand6 /@ #], 1] &,
    {{0, 0}}, n], 1]

inv6[M_, v_] := Module[{s =  $\sqrt{3}$ },
  {invSeg[M, {v, v + {2, 0}}],
    invSeg[M, {v + {2, 0}, v + {3, s}}],
    invSeg[M, {v + {3, s}, v + {2, 2 s}}],
invSeg[M, {v + {2, 2 s}, v + {0, 2 s}}],
    invSeg[M, {v + {0, 2 s}, v + {-1, s}}],
    invSeg[M, {v + {-1, s}, v}]}
]


```



```


With[{s =  $\sqrt{3}$ , M = {{1, 1/ $\sqrt{3}$ }, 1/ $\sqrt{3}$ }},
Graphics[{EdgeForm[Thin], Opacity[.5], Orange,
Map[Polygon[{#, # + {2, 0}, # + {3, s}, # +
{2, 2 s}, # + {0, 2 s}, # + {-1, s}]] &, ring6[2]], Red,
Disk@@M]}]

```




The second level of complexity corresponding to the above pattern is the following.

```
With[{M = {{1, 1/√3}, 1/√3}},  
Graphics[{Map[inv6[M, #] &, ring6[2]], Black, Circle@@M}]
```



This detail shows the sixth level of complexity, made with shapes formed with six arcs, all passing indirectly through q .

```
With[{M = {{1, 1/√3}, 1/√3}},
Graphics[Map[inv6[M, #] &, ring6[6]],
PlotRange → {{.8, 1.2}, {.47, .7}}]
```



■ An Inversive Spirograph


Finally, the following `Manipulate` shows an animated circle (orange) rotating inside a circle (pale brown) and the patterns generated by a point at the end of a line at a variable distance from the center of the circle. By varying the center and radius of the inversive circle, you can zoom in; the rotating radial line inverts into an arc orthogonal to the inversive path. You can enlarge the inversion by dragging the center of the inversive circle (light blue).

```
invCir[{q_, r_}, {Ac_, Ar_}] :=
Module[{n = disSq[Ac, q] - Ar^2},
Disk[q + (Ac - q) r^2 / n, Ar r^2 / Abs[n]]]
```

```

Manipulate[
Module[{M, cen, u},
cen = (1 - a) {Cos[t], Sin[t]};
M = {q, r};
Show[
Graphics[{EdgeForm[Thin], Opacity[.5],
If[w ≠ 2,
{ColorData[2, 5], Disk[], ColorData[2, 3],
Disk[cen, a], {Opacity[1], Black,
Line[{cen, u = cen + arm {Cos[t / a], -Sin[t / a]}]},
Disk[u, .04]}},
{}
],
If[w > 1,
{ColorData[2, 6], Disk[q, r], Red, Disk[q, .05],
ColorData[2, 5], invCir[M, {{0, 0}, 1]],
invCir[M, {cen, a}], Black,
{Opacity[1],
invSeg[M,
{cen, cen + arm {Cos[t / a], -Sin[t / a]}]}]},
{}
]}, PlotRange → 3],
If[w ≠ 2,
ParametricPlot[(1 - a) {Cos[u], Sin[u]} +
arm {Cos[u / a], -Sin[u / a]}, {u, 0, n π},
PlotPoints → 100],
{}
],
Quiet@If[w > 1,
ParametricPlot[
inver[M, (1 - a) {Cos[u], Sin[u]} +
arm {Cos[u / a], -Sin[u / a]}, {u, 0, n π},
PlotPoints → 300],
{}
]
]
],
{{a, 0.5, "radius of rotating circle"}, 0, 1,
Appearance → "Labeled"},
{{arm, 0.5, "length of rotating arm"}, .01, 2,
Appearance → "Labeled"},
{{q, {1, 2}}, Locator, Appearance → None},
Row[
{Control@{{r, 1, "radius of inversive circle"}, .01,
3, Appearance → "Labeled", ImageSize → Small},
Spacer[5], "Drag the red point."}],
{{n, 3, "number of turns"}, Range[10], Setter}, Delimiter,
{{w, 3, ""}, {1 → " spirograph only ",
2 → " inversion only ", 3 → " both "}},
{{t, 0, "rotate"}, 0, n π, Appearance → "Labeled"},
SaveDefinitions → True]

```



■ References

- [1] G. Smith, *A Mathematical Olympiad Primer*, London: United Kingdom Mathematics Trust, 2008.
- [2] A. Goucher. "Complex Projective 4-Space." (Jun 24, 2013). cp4space.wordpress.com/2012/11/04/final-chapter-of-mod4.
- [3] J. Rangel-Mondragon, "Inversive Patterns, Part I: Complex Inversion," *Mathematica in Education and Research*, **12**(2), 2007 pp. 162–183.
- [4] J. Rangel-Mondragon, "Inversive Patterns, Part II: Fundamental Properties of Inversion," *Mathematica in Education and Research*, **12**(4), 2007 pp. 330–354.
- [5] J. Rangel-Mondragon, "Inversive Patterns, Part III: Common Tangents, Mandalas and Gothic Windows," *Mathematica in Education and Research*, **12**(4), 2007 pp. 355–376.
- [6] N. J. A. Sloane. seq. A001189 in *The On-Line Encyclopedia of Integer Sequences*. oeis.org.
- [7] H. S. M. Coxeter and S. L. Greitzer, *Geometry Revisited*, New York: Random House, 1967.
- [8] E. W. Weisstein. "Nicomachus's Theorem" from Wolfram *MathWorld*—A Wolfram Web Resource. mathworld.wolfram.com/NicomachusTheorem.html.
- [9] M Schreiber. "A Visual Proof of Nicomachus's Theorem" from the Wolfram Demonstrations Project—A Wolfram Web Resource. demonstrations.wolfram.com/AVisualProofOfNicomachusTheorem.

J. Rangel-Mondragon, "Inversive Geometry: Part 1," *The Mathematica Journal*, 2013.
[dx.doi.org/doi:10.3888/tmj.15-7](https://doi.org/10.3888/tmj.15-7).

About the Author

Jaime Rangel-Mondragon received M.Sc. and Ph.D. degrees in applied mathematics and computation from the School of Mathematics and Computer Science at the University College of North Wales in Bangor, UK. He has been a visiting scholar at Wolfram Research, Inc. and has held positions in the Faculty of Informatics at UCNW, the Center for Linguistic and Literary Studies at the College of Mexico, the Department of Electrical Engineering at the Center for Research and Advanced Studies at the National Polytechnic Institute, the Center for Computational Engineering (of which he was director) at the Monterrey Institute of Technology, the Department of Mechatronics at the Queretaro Institute of Technology, and the Autonomous University of Queretaro in Mexico, where he is presently a member of the Faculty of Informatics and in charge of the academic unit of Algorithms, Computation, and Networks. His current research includes combinatorics, the theory of computing, computational geometry, and recreational mathematics.

Jaime Rangel-Mondragon
Autonomous University of Queretaro
Queretaro, Qro. Mexico
jrangeldragon@gmail.com